Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Isotropic Scattering

```
pIsotropic[u_] := 1/(4 Pi)
```

Normalization condition

```
Integrate[2 Pi pIsotropic[u], {u, -1, 1}]
1
```

Mean-cosine

```
Integrate[2 Pi pIsotropic[u] u, {u, -1, 1}]
0
```

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pIsotropic[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
1
```

```
Integrate[
 2 Pi (2 k + 1) pIsotropic[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
0
```

sampling

```
cdf = Integrate[2 Pi pIsotropic[u], {u, -1, x}]
(1 + x)/2
```

```
Solve[cdf == e, x]
{{x → -1 + 2 e}}
```

```
Clear[u]; Show[
 Plot[pIsotropic[u], {u, -1, 1}, PlotStyle → Thick]
 , Frame → True,
 FrameLabel → {{p[u],}, {"u = Cos[θ]", "Isotropic Scattering"}}]
```



Linearly-Anisotropic Scattering (Eddington)

```
pLinaniso[u_, b_] := 1/(4 Pi) (1 + b u)
```

```
Clear[u];
Show[
 Plot[pIsotropic[u], {u, -1, 1}, PlotStyle → Thick],
 Plot[pLinaniso[u, 1], {u, -1, 1}, PlotStyle → Dashed]
 , Frame → True,
 FrameLabel → {{p[u],}, {"u = Cos[θ]", "Linearly-Anisotropic Scattering"}}]
```



Normalization condition

```
Integrate[2 Pi pLinaniso[u, b], {u, -1, 1}, Assumptions → b > -1 && b < 1]
1
```

Mean cosine (g)

```
Integrate[2 Pi pLinaniso[u, b] u, {u, -1, 1}, Assumptions → b > -1 && b < 1]
```

$$\frac{b}{3}$$

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pLinaniso[Cos[y], b] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
```

1

```
Integrate[
 2 Pi (2 k + 1) pLinaniso[Cos[y], b] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
```

b

sampling

```
cdf = Integrate[2 Pi pLinaniso[u, b], {u, -1, x}]
```

$$\frac{1}{2} - \frac{b}{4} + \frac{x}{2} + \frac{b x^2}{4}$$

```
Solve[cdf == e, x]
```

$$\left\{\left\{x \to \frac{-1 - \sqrt{1 - 2 b + b^2 + 4 b e}}{b}\right\}, \left\{x \to \frac{-1 + \sqrt{1 - 2 b + b^2 + 4 b e}}{b}\right\}\right\}$$

```
b = 0.7;
Show[
 Plot[2 Pi pLinaniso[u, b], {u, -1, 1}],
 Histogram[
  Map[(-1 + Sqrt[1 - 2 b + b^2 + 4 b #])/b &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
]
Clear[b];
```



Rayleigh Scattering

General form:

```
pRayleigh[u_, γ_] := 1/(4 Pi) 3/(4 (1 + 2 γ)) ((1 + 3 γ) + (1 - γ) u^2)
```

Common special case ($\gamma = 0$):

```
pRayleigh[u_] := (1 + u^2) 3/(16 Pi)
```

Normalization condition

```
Integrate[2 Pi pRayleigh[u], {u, -1, 1}]
```

```
1
```

```
Integrate[2 Pi pRayleigh[u, y], {u, -1, 1}, Assumptions → y > 0] // Simplify
```

```
1
```

Mean cosine (g)

```
Integrate[2 Pi pRayleigh[u] u, {u, -1, 1}]
```

```
0
```

```
Integrate[2 Pi pRayleigh[u, y] u, {u, -1, 1}, Assumptions → y > 0] // Simplify
```

```
0
```

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
```

```
1
```

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
```

```
0
```

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 2, {y, 0, Pi}]
```

$\frac{1}{2}$

sampling

```
Show[
 Plot[2 Pi pRayleigh[u], {u, -1, 1}],
 Histogram[Map[(1 - (2 - 4 # + Sqrt[5 + 16 (-1 + #) #])^(2/3)) / (2 - 4 # + Sqrt[5 + 16 (-1 + #) #])^(1/3) &,
   Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
]
Clear[b];
```

$$\frac{1-\left(2-4\#+\sqrt{5+16\,(-1+\#)\,\#}\right)^{2/3}}{\left(2-4\#+\sqrt{5+16\,(-1+\#)\,\#}\right)^{1/3}}$$



Henyey-greenstein Scattering

```
Clear[pHG]; pHG[dot_, g_] := 1/(4 Pi) (1 - g^2)/(1 + g^2 - 2 g dot)^(3/2)
```

$$pHG[dot\_, g\_] := \frac{1}{4\,Pi}\,\frac{(1-g^2)}{(1+g^2-2\,g\,dot)^{\frac{3}{2}}}$$

```
pHGplot = Show[
  Plot[pHG[Cos[t], .8], {t, -Pi, Pi}, PlotRange → {0, 1}],
  Plot[pHG[Cos[t], .6], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .5], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .3], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],},
    {θ, "Henyey-Greenstein Scattering, g = 0.3, 0.4, 0.5, 0.6, 0.8"}}]
```



```
Show[
 ParametricPlot[{Sin[t], Cos[t]} (1),
  {t, -Pi, Pi}, PlotRange → All, PlotStyle → Dashed],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.75]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.68]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.6]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.5]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.3]),
  {t, -Pi, Pi}, PlotRange → All]
]
```



Normalization condition

```
Integrate[2 Pi pHG[u, g], {u, -1, 1}, Assumptions → g > -1 && g < 1]
1
```

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pHG[u, g] LegendreP[k, u] /. k → 0,
  {u, -1, 1}, Assumptions → g > -1 && g < 1]
```

1

```
Integrate[2 Pi (2 k + 1) pHG[u, g] LegendreP[k, u] /. k → 1,
  {u, -1, 1}, Assumptions → g > -1 && g < 1]
```

3 g

sampling

```
cdf = Integrate[2 Pi pHG[u, g], {u, -1, x}, Assumptions → g > -1 && g < 1 && x < 1]
```

$$\frac{(-1+g)\left(-1-g+\sqrt{1+g^2-2\,g\,x}\right)}{2\,g\,\sqrt{1+g^2-2\,g\,x}}$$

```
Solve[cdf == e, x]
```

$$\left\{\left\{x \to \frac{-1+2\,e+2\,g-2\,e\,g+2\,e^2\,g-g^2+2\,e\,g^2-2\,e\,g^3+2\,e^2\,g^3}{(1-g+2\,e\,g)^2}\right\}\right\}$$

```
FullSimplify[%]
```

$$\left\{\left\{x \to -\frac{(-1+g)^2+2\,e\,(-1+g)\,(1+g^2)-2\,e^2\,(g+g^3)}{(1+(-1+2\,e)\,g)^2}\right\}\right\}$$

```
g = 0.7;
Show[
  Plot[2 Pi pHG[u, g], {u, -1, 1}],
  Histogram[Map[-((-1 + g)^2 + 2 # (-1 + g) (1 + g^2) - 2 #^2 (g + g^3))/(1 + (-1 + 2 #) g)^2 &,
    Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
]
Clear[b, g];
```



Henyey-greenstein Scattering (Flatland)

Definition:

```
pH2[θ_, g_] := 1/(2 Pi) (1 - g^2)/(1 + g^2 - 2 g Cos[θ]);
```

$$pH2[\theta\_, g\_] := \frac{1}{2\,Pi}\,\frac{1-g^2}{1+g^2-2\,g\,Cos[\theta]};$$

Moments

```
Integrate[pH2[t, g] Cos[t], {t, -Pi, Pi}, Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g

```
Integrate[pH2[t, g] Cos[2 t], {t, -Pi, Pi},
 Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g^2

```
Integrate[pH2[t, g] Cos[7 t], {t, -Pi, Pi},
 Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g^7

Sampling:

```
g = -0.7;
Show[
 Histogram[Map[2 ArcTan[(1 - g)/(1 + g) Tan[Pi/2 (1 - 2 #)]] &,
   Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[ pH2[θ, g], {θ, -Pi, Pi}, PlotRange → All]
]
Clear[g];
```



Kagiwada-Kalaba (Ellipsoidal) Scattering

```
pEllipsoidal[u_, b_] := b (2 Pi Log[(1 + b) / (1 - b)] (1 - b u))^-1
```

```
pEllplot = Show[
  Plot[pEllipsoidal[Cos[t], .9], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .8], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .65], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .95], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],}, 
    {θ, "Ellipsoidal Scattering, b = 0.4, 0.65, 0.8, 0.9, 0.95"}}]
```



sampling

```
b = -0.8;
Show[Histogram[
  Map[(1 - (1 + b) ((1+b)/(1-b))^-#)/b &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[2 Pi pEllipsoidal[u, b], {u, -1, 1}]

]
Clear[b];
```



Binomial Scattering

```
pBinomial[u_, n_] := Pi^-1 ((n + 1) / 2^(n+2)) (1 + u)^n
```

```
pBinplot = Show[
  Plot[pBinomial[Cos[t], 1], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 2], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 3], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 5], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],}, {θ, "Binomial Scattering, n = 1, 2, 3, 4, 5"}}]
```



Normalization condition

```
Integrate[2 Pi pBinomial[u, n], {u, -1, 1}, Assumptions → n ≥ 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi pBinomial[u, n] u, {u, -1, 1}, Assumptions → n ≥ 0]
```

$$\frac{n}{2+n}$$

sampling

```
n = 25.8;
Show[
 Histogram[Map[-1 + (2^(1+n) #)^(1/(1+n)) &, Table[RandomReal[], {i, 1, 100000}]], 50, "PDF"],
 Plot[2 Pi pBinomial[u, n], {u, -1, 1}, PlotRange → All]

]
Clear[b];
```



Liu Scattering

```
pLiu[u_, e_, m_] := e (2 m + 1) (1 + e u)^(2 m) / (2 Pi ((1 + e)^(2 m+1) - (1 - e)^(2 m+1)))

Clear[m]
```

```
pLiuplot = Show[
  Plot[pLiu[Cos[t], 4, 2], {t, -Pi, Pi}, PlotRange → All],
  Plot[pLiu[Cos[t], 7, 2], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel →
   {{p[Cos[θ]],}, {θ, "Liu Scattering, (m = 2, ϵ = 4), (m = 2, ϵ = 7)"}}]
```



Normalization condition

```
Integrate[2 Pi pLiu[u, e, m], {u, -1, 1}, Assumptions → e > 0 && m > 0 && m ∈ Integers]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pLiu[u, e, m], {u, -1, 1},
 Assumptions → e > 0 && m > 0 && m ∈ Integers && e < 1]
```

$$\frac{(1+e)^{1+2m}\left(-1+e+2em\right)+(1-e)^{1+2m}\left(1+e+2em\right)}{2e\left(-(1-e)^{1+2m}+(1+e)^{1+2m}\right)(1+m)}$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pLiu[u, e, m] LegendreP[k, u] /. k → 0, {u, -1, 1},
 Assumptions → m > 0 && m ∈ Integers && e ∈ Reals && e ≠ 0 && Abs[e] < 1]
```

1

```
Integrate[2 Pi (2 k + 1) pLiu[u, e, m] LegendreP[k, u] /. k → 2, {u, -1, 1},
 Assumptions → m > 0 && m ∈ Integers && e ∈ Reals && e ≠ 0 && Abs[e] < 1]
```

$$\left(5\left((1+e)^{1+2m}\left(3+e\left(-3+2m\left(-3+2e(1+m)\right)\right)\right)+(1-e)^{2m}(-1+e)\left(3+e\left(3+2m\left(3+2e(1+m)\right)\right)\right)\right)\right)\Big/\left(2e^{2}\left(-(1-e)^{1+2m}+(1+e)^{1+2m}\right)(1+m)(3+2m)\right)$$

sampling

```
m = 3.5;
ϵ = 0.9;
```

$$\text{Show}\Big[\text{Histogram}\Big[\text{Map}\Big[\frac{-1+\left((-1+\#)\ (1-\epsilon)^{2\,m}\ (-1+\epsilon)+\#\ (1+\epsilon)^{1+2\,m}\right)^{\frac{1}{1+2\,m}}}{\epsilon}\ \&,$$

```
    Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
  Plot[2 Pi pLiu[u, ϵ, m], {u, -1, 1}, PlotRange → All]

]
Clear[m, ϵ];
```



Gegenbauer Scattering

$$\text{pGegenbauer}[u\_,\ g\_,\ a\_]:=\frac{\left(1+g^{2}-2\,g\,u\right)^{-(a+1)}}{\frac{\left((1-g)^{-2\,a}-(1+g)^{-2\,a}\right)\,\pi}{a\,g}}$$

```
Show[
 Plot[pGegenbauer[Cos[t], 0.5, 1], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 3], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 5], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel →
  {{p[Cos[θ]],}, {θ, "Gegenbauer Scattering, G = 0.5, a = 1, 3, 5"}}]
```



Normalization condition

```
Integrate[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

$$\frac{(1+g)^{2a}\left(1-2ag+g^2\right)-(1-g)^{2a}\left(1+2ag+g^2\right)}{2(-1+a)g\left((1-g)^{2a}-(1+g)^{2a}\right)}$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 0,
 {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

1

```
FullSimplify[Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 3,
  {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]]
```

$$-\left(7\left(24a^2g^2\left(1+g^2\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+3\left(5+3g^2+3g^4+5g^6\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+8a^3g^3\left((1-g)^{2a}+(1+g)^{2a}\right)+2ag\left(15+14g^2+15g^4\right)\left((1-g)^{2a}+(1+g)^{2a}\right)\right)\right)/\left(8(-3+a)(-2+a)(-1+a)g^3\left((1-g)^{2a}-(1+g)^{2a}\right)\right)$$

sampling

```
g = -0.8;
a = -1.2;
Show[Histogram[Map[(1 + g^2 - (# (1 - g)^(-2 a) - (-1 + #) (1 + g)^(-2 a))^(-1/a))/(2 g) &,
    Table[RandomReal[], {i, 1, 100 000}]], 100, "PDF"],
 Plot[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, PlotRange → All]

]
Clear[g, a];
```



vMF (spherical Gaussian) Scattering

```
pVMF[u_, k_] := k/(4 Pi Sinh[k]) Exp[k u]
```

```
Show[
 Plot[pVMF[Cos[t], 5.8], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 15], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 30], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel → {{p[Cos[θ]],}, {θ, "vMF, k = {5.8,15,30}"}}]
```



Normalization condition

```
Integrate[2 Pi pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

$$-\frac{1}{k} + \text{Coth}[k]$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 4,
 {u, -1, 1}, Assumptions → k > 0]
```

$$\frac{9\left(105 + 45\,k^2 + k^4 - 5\,k\left(21 + 2\,k^2\right)\text{Coth}[k]\right)}{k^4}$$

sampling

```
k = 3;
Show[Histogram[
  Map[Log[E^-k (1 - #) + E^k #]/k &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[2 Pi pVMF[u, k], {u, -1, 1}, PlotRange → All]

]
Clear[k];
```



Klein-Nishina

Normalized variant of Klein-Nishina - energy parameter “e” = $\frac{E_\gamma}{m_e c^2}$

```
pKleinNishina[u_, e_] := 1/(1 + e (1 - u)) 1/((2 π Log[1+2 e])/e)
```

Normalization condition

In[•]:= `Integrate[2 Pi pKleinNishina[u, e], {u, -1, 1}, Assumptions → e > 0]`

Out[•]= 1

Mean-cosine

In[•]:= `Integrate[2 Pi pKleinNishina[u, e] u, {u, -1, 1}, Assumptions → e > 0]`

Out[•]= $1 + \frac{1}{e} - \frac{2}{\text{Log}[1 + 2 e]}$

Legendre expansion coefficients

In[•]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[•]= 1

In[•]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[•]= $3 + \frac{3}{e} - \frac{6}{\text{Log}[1 + 2\,e]}$

In[•]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[•]= $\frac{5}{4}\left(1 + \frac{3\left(2 + 4\,e + e^2 - \frac{4\,e\,(1+e)}{\text{Log}[1+2\,e]}\right)}{e^2}\right)$

In[•]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[•]= $\frac{7\left(15 + 45\,e + 36\,e^2 + 6\,e^3 - \frac{2\,e\,(15+30\,e+11\,e^2)}{\text{Log}[1+2\,e]}\right)}{6\,e^3}$

sampling

In[•]:=
```
cdf = Integrate[2 Pi pKleinNishina[u, e], {u, -1, x}, Assumptions → e > 0 && 0 < x < 1]
```

Out[•]= $1 - \frac{\text{Log}[1 + e - e\,x]}{\text{Log}[1 + 2\,e]}$

In[•]:=
```
Solve[cdf == k, x]
```

Out[•]= $\left\{\left\{x \to \text{ConditionalExpression}\left[\frac{1 + e - (1 + 2\,e)^{1-k}}{e}, -\pi \le \text{Im}\left[(-1 + k)\,\text{Log}[1 + 2\,e]\right] < \pi\right]\right\}\right\}$

In[•]:=
```
With[{e = 1.1},

 Show[
  Plot[2 Pi pKleinNishina[u, e], {u, -1, 1}],
  Histogram[
   Map[(1 + e - (1 + 2 e)^(1-#))/e &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
  ]
 ]
```

Out[•]=



Cornette-Shanks

[Cornette and Shanks 1992] - *Physically reasonable analytic expression for the single-scattering phase function.*

Independently proposed [Liu and Weng 2006]

In[1016]:=
$$\text{pCornetteShanks}[u\_, g\_] := \frac{3}{8\,\text{Pi}} \frac{\left(1-g^2\right)\left(1+u^2\right)}{\left(2+g^2\right)\left(1+g^2-2\,g\,u\right)^{3/2}}$$

Normalization condition

In[1018]:=
```
Integrate[2 Pi pCornetteShanks[u, g], {u, -1, 1}, Assumptions → -1 < g < 1]
```

Out[1018]= 1

Mean-cosine

In[1019]:=
```
Integrate[2 Pi pCornetteShanks[u, g] u, {u, -1, 1}, Assumptions → -1 < g < 1]
```

Out[1019]=
$$\frac{3\,g\,\left(4+g^2\right)}{5\,\left(2+g^2\right)}$$

Legendre expansion coefficients

In[1023]:=
```
Integrate[
 2 Pi (2 k + 1) pCornetteShanks[Cos[y], g] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
 {y, 0, Pi}, Assumptions → -1 < g < 1]
```

Out[1023]= 1

In[1024]:=
```
Integrate[
 2 Pi (2 k + 1) pCornetteShanks[Cos[y], g] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
 {y, 0, Pi}, Assumptions → -1 < g < 1]
```

Out[1024]= $$\frac{9\, g\, \left(4 + g^2\right)}{5\, \left(2 + g^2\right)}$$

In[1025]:=
```
Integrate[
 2 Pi (2 k + 1) pCornetteShanks[Cos[y], g] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
 {y, 0, Pi}, Assumptions → -1 < g < 1]
```

Out[1025]= $$\frac{7 + 80\, g^2 + 18\, g^4}{14 + 7\, g^2}$$

In[1026]:=
```
Integrate[
 2 Pi (2 k + 1) pCornetteShanks[Cos[y], g] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
 {y, 0, Pi}, Assumptions → -1 < g < 1]
```

Out[1026]= $$\frac{g\, \left(27 + 238\, g^2 + 50\, g^4\right)}{15\, \left(2 + g^2\right)}$$

sampling

In[1027]:=
```
cdf = Integrate[2 Pi pCornetteShanks[u, g],
  {u, -1, x}, Assumptions → -1 < g < 1 && 0 < x < 1]
```

Out[1027]=
$$\frac{1}{4\, g^3\, \left(2 + g^2\right)\, \sqrt{1 + g^2 - 2\, g\, x}} \left(2 - 2\, g^6 - 2\, g\, x - 2\, \sqrt{1 + g^2 - 2\, g\, x} + 4\, g^3\, \sqrt{1 + g^2 - 2\, g\, x} + g^4\, \left(-5 + x^2\right) + 2\, g^5\, \left(x + \sqrt{1 + g^2 - 2\, g\, x}\right) - g^2\, \left(-5 + x^2 + 4\, \sqrt{1 + g^2 - 2\, g\, x}\right)\right)$$

Draine

Draine, B.T. (2003) '*Scattering by interstellar dust grains. 1: Optical and ultraviolet*', ApJ., 598, 1017–25.

In[1057]:=
$$\texttt{pDraine[u\_, g\_, α\_] :=}\ \frac{1}{4\,\mathrm{Pi}} \left(\frac{1 - g^2}{\left(1 + g^2 - 2\, g\, u\right)^{3/2}}\ \frac{1 + \alpha\, u^2}{1 + \alpha\, \left(1 + 2\, g^2\right)/3}\right)$$

Normalization condition

In[1058]:=
```
Integrate[2 Pi pDraine[u, g, a], {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1058]= 1

Mean-cosine

In[1059]:=
```
Integrate[2 Pi pDraine[u, g, a] u, {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1059]= $\frac{3}{5}\left(g+\frac{2\ (1+a)\ g}{3+a+2\ a\ g^2}\right)$

In[1060]:= $\frac{3}{5}\left(g+\frac{2\ (1+a)\ g}{3+a+2\ a\ g^2}\right)$ /. a → 0

Out[1060]= g

Legendre expansion coefficients

In[1064]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1064]= 1

In[1065]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1065]= $\frac{9\ g\ \left(5+a\ \left(3+2\ g^2\right)\right)}{5\ \left(3+a+2\ a\ g^2\right)}$

In[1066]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1066]= $\frac{14\ a+5\ (21+11\ a)\ g^2+36\ a\ g^4}{7\ \left(3+a+2\ a\ g^2\right)}$

In[1067]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```
Out[1067]= $\frac{g\ \left(54\ a+7\ (45+23\ a)\ g^2+100\ a\ g^4\right)}{15\ \left(3+a+2\ a\ g^2\right)}$

sampling

In[1070]:=
```
cdf = Integrate[2 Pi pDraine[u, g, a],
  {u, -1, x}, Assumptions → 0 < a < 1 && -1 < g < 1 && -1 < x < 1]
```

Out[1070]=
$$\left(3\,(-1+g)\,g^2\left(-1-g+\sqrt{1+g^2-2\,g\,x}\right)+a\left(2-2\,g^6-2\,g\,x-2\sqrt{1+g^2-2\,g\,x}+g^3\sqrt{1+g^2-2\,g\,x}+g^4\left(-2+x^2\right)+2\,g^5\left(x+\sqrt{1+g^2-2\,g\,x}\right)-g^2\left(-2+x^2+\sqrt{1+g^2-2\,g\,x}\right)\right)\right)\Big/\left(2\,g^3\left(3+a+2\,a\,g^2\right)\sqrt{1+g^2-2\,g\,x}\right)$$

Schlick

In[1079]:=
$$\text{pSchlick}[u\_,\,k\_] := \frac{1}{4\,\text{Pi}}\left(\frac{1-k^2}{(1+k\,u)^2}\right)$$

Normalization condition

In[1080]:=
```
Integrate[2 Pi pSchlick[u, k], {u, -1, 1}, Assumptions → -1 < k < 1]
```

Out[1080]= 1

Mean-cosine

In[1081]:=
```
Integrate[2 Pi pSchlick[u, k] u, {u, -1, 1}, Assumptions → -1 < k < 1]
```

Out[1081]=
$$-\frac{k-\text{ArcTanh}[k]+k^2\,\text{ArcTanh}[k]}{k^2}$$

Legendre expansion coefficients

In[1082]:=
```
Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
  {y, 0, Pi}, Assumptions → -1 < e < 1]
```

Out[1082]= ConditionalExpression[1, e ≠ 0]

In[1083]:=
```
Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
  {y, 0, Pi}, Assumptions → -1 < e < 1]
```

Out[1083]=
$$\text{ConditionalExpression}\left[-\frac{3\left(e+\left(-1+e^2\right)\text{ArcTanh}[e]\right)}{e^2},\;e\neq 0\right]$$

In[1084]:=
```
Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
  {y, 0, Pi}, Assumptions → -1 < e < 1]
```

Out[1084]=
$$\text{ConditionalExpression}\left[-\frac{5\left(-6\,e+4\,e^3-6\left(-1+e^2\right)\text{ArcTanh}[e]\right)}{2\,e^3},\;e\neq 0\right]$$

In[1085]:=
```
Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}, Assumptions → -1 < e < 1]
```

Out[1085]= ConditionalExpression$\left[-\frac{7\left(30\,e-26\,e^3-6\left(5-6\,e^2+e^4\right)\mathrm{ArcTanh}[e]\right)}{4\,e^4},\ e\neq 0\right]$

sampling

In[1086]:=
```
cdf = Integrate[2 Pi pSchlick[u, e], {u, -1, x}, Assumptions → -1 < e < 1 && 0 < x < 1]
```

Out[1086]= $\frac{(1+e)\ (1+x)}{2+2\,e\,x}$

In[1087]:=
```
Solve[cdf == k, x]
```

Out[1087]= $\left\{\left\{x\to\frac{1+e-2\,k}{-1-e+2\,e\,k}\right\}\right\}$

In[1089]:=
```
With[{e = -.7},
 Show[
  Plot[2 Pi pSchlick[u, e], {u, -1, 1}],
  Histogram[Map[(1 + e - 2 #)/(-1 - e + 2 e #) &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
 ]
]
```

Out[1089]=